

Elisabeth Fughe  
Matrikelnummer: 5263769  
s3499227@stud.uni-frankfurt.de

**Bachelorarbeit (B.Sc. - Informatik)**

# **Können Modelle der Ökonophysik Preis-Dynamiken vorhersagen?**

**Fitting von Markt-Modellen auf empirischen Daten mittels Bayesscher  
Statistik**

Elisabeth Fughe

Abgabedatum: 16. April 2019

FIAS - Frankfurt Institute for Advanced Studies  
Prof. Dr. Nils Bertschinger

## **Zusammenfassung**

tbd

# Inhaltsverzeichnis

<b>1. Einleitung</b>	<b>5</b>
<b>2. Verwendete Methoden</b>	<b>6</b>
2.1. Bayessche Modellierung . . . . .	6
2.2. Markov Chain Monte Carlo (MCMC) . . . . .	7
2.3. Hamiltonian Monte Carlo Sampling (HMC) . . . . .	10
2.4. <i>Stan</i> und <i>R</i> . . . . .	13
<b>3. Die Modelle</b>	<b>17</b>
3.1. Generalized Autoregressive Conditional Heteroscedasticity (GARCH) . . . . .	17
3.2. Franke & Westerhoff (FW) . . . . .	20
3.3. AL herd walk (AL) . . . . .	23
<b>4. Simulationen</b>	<b>24</b>
4.1. Daten & Vorhersagen . . . . .	24
4.2. Ergebnisse . . . . .	24
<b>Anhang</b>	<b>26</b>
A. <i>Stan</i> Code . . . . .	26
B. <i>R</i> Code . . . . .	31
<b>Literatur</b>	<b>32</b>
<b>Eigenständigkeitserklärung</b>	<b>33</b>

## Abbildungsverzeichnis

2.1.	Verhalten der Markov-Kette beim Ziehen von Stichproben [2] . . . . .	7
2.2.	Konzentration der Markov-Kette auf die typische Menge einer Verteilung [2]	8
2.3.	Starke Krümmungen (grün) der typischen Menge einer Verteilung führen zu verzerrten MCMC-Schätzern [2] . . . . .	9
2.4.	Mit Hilfe eines Vektorfeldes das an der typischen Menge ausgerichtet ist, kann die Ziel-Verteilung leichter abgetastet werden [2]. . . . .	10
2.5.	Die Richtung des Gradienten zeigt immer von der typischen Menge in Richtung parameter-sensitiver Bereiche(grüner Pfeil), wie z.B. dem Modus der Verteilung. Folgt der Sampler diesen Richtungen, kann er die typische Menge nicht erkunden [2]. . . . .	11
2.6.	Analoges System der klassischen Physik [2]. . . . .	12
3.1.	Die Daten weisen im Zeitverlauf unterschiedliche Abweichung zur Trendgerade auf, d.h. die Residuen/Störterme sind nicht konstant im Zeitverlauf und es liegt Heteroskedastizität vor (Eigene Darstellung). . . . .	18

# 1. Einleitung

tbd

## 2. Verwendete Methoden

In den nachfolgenden Kapiteln 2.1 bis 2.4 werden die mathematischen Grundlagen, die verwendeten Methoden sowie die Programmiersprache erläutert. Um verschiedene Modelle auf den gleichen Daten anzuwenden und vergleichen zu können, wurden die Modelle von Bertschinger, Mozzhorin und Sinha [1] in der Programmiersprache *Stan* beschrieben. Anschließend ist das Fitten der Modelle auf den Daten mittels **R** leicht umsetzbar.

### 2.1. Bayessche Modellierung

Die Bayessche Statistik untersucht mittels Bayesscher Wahrscheinlichkeiten und dem Satz von Bayes Fragestellungen der Stochastik. Anders als in der klassischen Statistik, die unendlich oft wiederholbare Zufallsexperimente voraussetzt, steht die Verwendung und Modellierung von Wahrscheinlichkeitsverteilungen im Vordergrund.

Es gilt, dass beobachtete Daten

$$x = (x_1, \dots, x_n)$$

mittels bedingter Wahrscheinlichkeiten in Beziehung zu unbekannten Parametern

$$\theta = (\theta_1, \dots, \theta_m)$$

stehen. So kann die gemeinsame Wahrscheinlichkeitsdichte

$$p(x, \theta) = p(x|\theta) \cdot p(\theta)$$

durch die a-priori-Verteilung unbekannter Parameter  $p(\theta)$  und den Erkenntnissen aus dem Datensatz  $p(x|\theta)$  berechnet werden. Mit Hilfe des Satzes von Bayes kann dann die a-posteriori-Verteilung unbekannter Parameter

$$p(\theta|x) = \frac{p(x|\theta) \cdot p(\theta)}{p(x)}$$

ermittelt werden [1].

Die a-posteriori-Verteilung enthält somit Informationen über die unbekannten Parameter durch die Kombination der a-priori Verteilung mit den Informationen, die aus den beobachteten Daten gewonnen wurden. Sie wird zur Punktschätzung und zur Schätzung von Konfidenzintervallen genutzt und daher auch oft als Ziel-Verteilung bezeichnet.

So sind Bayessche Modelle im Gegensatz zur klassischen Statistik auf kleineren Datensätzen anwendbar, dort ergibt sich jedoch eine breite Wahrscheinlichkeitsverteilung, die somit unter Umständen eine geringe Genauigkeit aufweist.

## 2.2. Markov Chain Monte Carlo (MCMC)

In der Bayesschen Statistik beschreibt die a-posteriori-Verteilung die Unsicherheit der unbekannten Parameter, die anhand beobachteter Daten geschätzt wurden. Mit der Markov Chain Monte Carlo (MCMC) Methode können die a-posteriori-Verteilung und dadurch die unbekannten Parameter untersucht werden [8].

Um aus der Ziel-Verteilung Stichproben ziehen zu können, wird eine Markov-Kette entworfen, deren stationäre Verteilung der Ziel-Verteilung entspricht. Hier wird eine Stichprobe aus der a-posteriori-Wahrscheinlichkeitsdichte benötigt, sodass das langfristige Gleichgewicht der Markov-Kette der a-posteriori-Wahrscheinlichkeitsdichte entspricht. Der Zustand der Markov-Kette, der sich möglichst nah am Gleichgewicht befindet, entspricht dann einem Element der Stichprobe. Es gilt somit, dass sich die Qualität der Stichprobe mit der Anzahl der Schritte, die die Markov-Kette zurücklegt, verbessert.

Zur Konstruktion der Markov-Kette wird eine Übergangsfunktion berechnet, die die Ziel-Wahrscheinlichkeitsdichte invariant lässt, damit die Ziel-Dichte der stationären Verteilung der Markov-Kette entspricht. Sei z.B.  $p(\theta)$  die Ziel-Dichte und sei  $t(\theta'|\theta)$  die Übergangsdichte, dann gilt

$$p(\theta') = \int t(\theta'|\theta) \cdot p(\theta) d\theta$$

Stellt man sich die Markov-Kette als zufällig durch einen Parameterraum, also z.B. die a-posteriori-Verteilung, wandernden Punkt vor, so entspricht die  $n$ -te Komponente der Verteilung der relative Wahrscheinlichkeit, den Punkt im Zustand  $n$  anzutreffen.




Abbildung 2.1.: Verhalten der Markov-Kette beim Ziehen von Stichproben [2]

Die Übergangsdichte (grün) (a) in Abbildung 2.1 beschreibt die Wahrscheinlichkeit eines neuen Punktes im Parameterraum in Abhängigkeit der aktuellen Position d.h. die Wahrscheinlichkeit des nächsten Schrittes der Markov-Kette auf ihrem Weg durch die Verteilung in Abhängigkeit ihrer aktuellen Position. Werden Stichproben aus dieser Verteilung gezogen, also wurden zufällig gewählte nächste Schritte vom Algorithmus akzeptiert, ergibt sich ein weiterer Zustand in der Markov-Kette und eine neue Verteilung, von der Stichproben ge-

zogen werden können (vgl. (b) in Abbildung 2.1). So wandert die Markov-Kette durch den Parameterraum (c) (vgl. Abbildung 2.1).




Abbildung 2.2.: Konzentration der Markov-Kette auf die typische Menge einer Verteilung [2]

Wenn die Übergangsdichte die Ziel-Verteilung beibehalten soll, konzentriert sie sich auf deren typische Menge<sup>1</sup> (rot) siehe Abbildung 2.2. Unter idealen Bedingungen entdeckt die Markov-Kette in drei verschiedenen Phasen:

1. Die Markov-Kette konvergiert gegen die typische Menge der Verteilung. Die MCMC-Schätzer sind stark verzerrt.
2. Die Markov-Kette hat die typische Menge erkannt und verweilt dort. Diese erste Erkundung der typischen Menge ist äußerst effektiv und verringert die Verzerrung der MCMC-Schätzer erheblich, da die Verzerrung der ersten Stichproben eliminiert wird.
3. In der dritten Phase wird die Präzision der MCMC-Schätzer durch weiteres Erforschen der typischen Menge weiter verfeinert, so dass sie den zentralen Grenzwertsatz erfüllen. Das bedeutet, dass die durch die Markov-Kette gezogenen Stichproben asymptotisch der Ziel-Verteilung folgen.

Ideale Bedingungen liegen allerdings nicht vor, wenn die typische Menge der a-posteriori-Verteilung z.B. eine starke Krümmung aufweist (siehe Abbildung 2.3). Die Markov-Kette kann damit nicht umgehen und aus diesem stark gekrümmten Bereich keine Stichprobe ziehen und ignoriert diesen einfach und verzerrt dadurch die MCMC-Schätzer. Da Markov-Ketten die exakten Erwartungen asymptotisch wiederherstellen müssen, ist eine Kompensation dafür nötig, dass manche Regionen der Ziel-Verteilung nicht beachtet wurden. Dazu bleibt die Markov-Kette an der Grenze der Krümmung hängen. So werden die MCMC-Schätzer annähernd so berechnet, als würde die Markov-Kette den gekrümmten Bereich erkunden. Allerdings erhält man so überschätzte MCMC-Schätzer. Eine starke Krümmung in der typischen Menge der Ziel-Verteilung führt also zu verzerrten MCMC-Schätzern, die den zentralen Grenzwertsatz nicht mehr erfüllen [2].

<sup>1</sup>**Typical Set (hier: typische Menge):** Die typische Menge ist ein Begriff aus der Informationstheorie und hängt direkt mit dem Begriff der Entropie zusammen. Es gilt, dass die Summe aller Wahrscheinlichkeiten aller Elemente der typischen Menge annähernd 1 ist [11]. Daher enthält die typische Menge eine gute Beschreibung des Parameterraums.





Abbildung 2.3.: Starke Krümmungen (grün) der typischen Menge einer Verteilung führen zu verzerrten MCMC-Schätzern [2]

Ein weit verbreitetes MCMC-Verfahren zur Konzentration auf die typische Menge der zu untersuchenden Verteilung ist der Metropolis-Hastings-Algorithmus. Der Algorithmus startet an einem zufälligen Punkt in der zu untersuchenden Verteilung. Dann wird eine Schrittweite zufällig mit Hilfe einer symmetrischen Wahrscheinlichkeitsverteilung gewählt. Der Schritt wird auf Grundlage der Wahrscheinlichkeit der neuen Position im Verhältnis zur alten Position abgelehnt oder akzeptiert [8]. So wird sicher gestellt, dass die Markov-Kette von jedem Punkt aus gegen das langfristige Gleichgewicht der Ziel-Wahrscheinlichkeitsdichte konvergiert [1].

Der Metropolis-Hastings-Algorithmus ist sehr einfach zu implementieren und liefert gute Ergebnisse, allerdings nicht bei stark korrelierten Parametern [8] und hoher Komplexität der Verteilung [2]. Denn wenn die a-posteriori-Verteilung stark gekrümmt ist, bleibt die Markov-Kette an diesen Rändern hängen und lehnt so gut wie jeden weiteren Schritt ab. So kann nicht aus allen Bereiche der Ziel-Verteilung Stichproben gezogen werden und es entsteht eine verzerrte Stichprobe.

Je mehr Dimensionen die Verteilung hat, desto kleiner wird die typische Menge der Verteilung. Das führt dazu, dass fast immer Punkte außerhalb der typischen Menge gewählt werden. Die Dichte des Punktes ist so klein, dass der Schritt vom Metropolis-Hastings-Algorithmus immer abgelehnt wird. Das führt dazu, dass sich die Markov-Kette kaum weiter bewegt. Erhöht man die Akzeptanz, indem man die Anzahl der Punkte reduziert, die sich in der typischen Menge befinden müssen, führt das dazu, dass die Markov-Kette extrem langsam konvergiert [2]. In der Theorie würde die Markov-Kette das langfristige Gleichgewicht erreichen, aber in der Praxis stehen dazu nicht unendliche Ressourcen zur Berechnung zur Verfügung.

Der Metropolis-Hastings-Algorithmus kann auch als stochastisches Optimierungsverfahren verstanden werden. Mittels Simulated Annealing<sup>2</sup> nähert man sich mit immer höherer Wahrscheinlichkeit an ein Minimum an. Ist die Krümmung der Verteilung zu stark, kann der Algorithmus das lokale Minimum nicht mehr verlassen und so wird fast jeder weitere Schritt der

<sup>2</sup>**Simulated Annealing:** „Ein heuristisches Approximationsverfahren. Es wird zum Auffinden einer Näherungslösung von Optimierungsproblemen eingesetzt, die durch ihre hohe Komplexität das vollständige Ausprobieren aller Möglichkeiten und mathematische Optimierungsverfahren ausschließen“[10].

Markov-Kette abgelehnt. Letztlich wird der Bereich der Verteilung ignoriert bzw. die Markov-Kette bleibt am Rand des unerwarteten Bereichs hängen. Das führt zu einer Verzerrung der resultierenden MCMC-Schätzer [2].

### 2.3. Hamiltonian Monte Carlo Sampling (HMC)

In Modellen mit vielen Dimensionen ist es möglich, dass die Markov-Kette mittels Metropolis-Hastings-Algorithmus nicht alle Bereiche der Ziel-Verteilung in einer angemessenen Zeit abtasten kann. Die Strategie des Metropolis-Algorithmus, zufällige Schritte zu wählen und danach zu entscheiden, ob der Schritt akzeptiert wird oder nicht, ist für Modelle mit vielen Dimensionen weniger erfolgreich. Denn es existieren exponentiell viele Richtungen, in die sich der Sampler bewegen kann, aber nur ein Bruchteil der Richtungen belässt den Sampler in der typischen Menge und führt letztlich zu akzeptierten Stichproben.

Um große Schritte vom initialen Punkt in unentdeckte Bereiche der Ziel-Verteilung machen zu können, müssen Informationen über die Geometrie der Verteilung genutzt werden. Die Markov-Kette soll Übergänge berechnen, die der Masse mit hohen Wahrscheinlichkeiten folgt, sich also zusammenhängend durch die typische Menge bewegen. Diese Hamiltonian-Markov-Übergänge können durch Nutzung der differenzierbaren Struktur der a-posteriori-Verteilung berechnet werden.




Abbildung 2.4.: Mit Hilfe eines Vektorfeldes das an der typischen Menge ausgerichtet ist, kann die Ziel-Verteilung leichter abgetastet werden [2].

Um die Zielverteilung schneller und gezielter abtasten zu können, wird jedem Punkt im Parameterraum eine Richtung für eine kleine Distanz zugewiesen, z.B. durch ein Vektorfeld, das an der typischen Menge ausgerichtet ist (vgl. Abbildung 2.4). Folgt der Sampler den Richtungen, entsteht eine zusammenhängende Kurve (*engl. Trajectory*), die effizient und

möglichst schnell vom initialen Punkt durch unentdeckte Bereiche der typischer Menge der Ziel-Verteilung führt [2].

Um jedem Punkt eine Richtung zuzuweisen, kann ein Vektorraum mittels Informationen der Ziel-Verteilung konstruiert werden. Dazu wird die differenzielle geometrische Struktur der a-posteriori-Verteilung benötigt, die wir durch den Gradienten der Ziel-Dichte erhalten können. Denn der Gradient der Ziel-Dichte definiert eben ein solches für die geometrische Struktur der Verteilung sensitives Vektorfeld im Parameterraum.




Abbildung 2.5.: Die Richtung des Gradienten zeigt immer von der typischen Menge in Richtung parameter-sensitiver Bereiche(grüner Pfeil), wie z.B. dem Modus der Verteilung. Folgt der Sampler diesen Richtungen, kann er die typische Menge nicht erkunden [2].

Allerdings hängen Ziel-Dichte sowie ihr Gradient stark von den Verteilungsparametern ab, sodass der Gradient immer in Richtung parameter-sensitiver Bereiche wie z.B. dem Modus<sup>3</sup> der Verteilung zeigt (vgl. Abbildung 2.5). Das Vektorfeld enthält somit nicht genügend Informationen, um den Sampler durch die typische Menge in parameter-invariante Bereiche zu führen. Um parameter-invariante Bereiche der Ziel-Dichte zu erkunden, muss der Gradient um zusätzliche geometrische Informationen ergänzt werden.

Das Ergänzen der Informationen ist durch Differentialgeometrie möglich, die auch die Mathematik der klassischen Physik beschreibt. Es gilt, dass für jedes probabilistische System ein mathematisch äquivalentes physikalisches System existiert [2]. Das Erkunden der Ziel-Dichte kann äquivalent auch als Erkunden eines physikalischen Systems beschrieben werden. Sei der Modus der Ziel-Dichte ein Planet und der Gradient sein Gravitationsfeld, dann sei die typische Menge der Dichte, der Bereich um den Planeten, durch den ein Satellit kreisen soll (vgl.

<sup>3</sup>Modus: Der Modus ist ein Begriff der deskriptiven Statistik. Er entspricht dem häufigsten Wert einer Stichprobe [9].

Abbildung 2.6).




Abbildung 2.6.: Analoges System der klassischen Physik [2].

Es gilt also das probabilistische Modell so um Impulse (*engl. Momentum*) zu erweitern, dass analog der Satellit die Anziehungskraft des Planeten ausbalanciert und die gewünschte Umlaufbahn nicht verlässt [2]. Der Hamiltonian Monte Carlo Algorithmus ist der einzige Algorithmus, der Impulse mit einer probabilistischen Struktur, die konservative Dynamiken erlaubt, implementiert [2].

Der HMC-Algorithmus zieht also Stichproben aus einem erweiterten Zustandsraum mit Impuls  $i$  eines Teilchens an der Position  $P$ .

Für die Dichte  $d(P, i)$  des Zustandsraums gilt [1]:

$$\begin{aligned} d(P, i) &= d(P)d(i|P) \\ &= e^{\log d(P) + \log d(i|P)} \\ &= e^{-H(P, i)} \end{aligned} \tag{2.1}$$

Wobei die Hamiltonian-Korrektur  $H(P, i)$  in Analogie zum physikalischen System der Summe potentieller und kinetischer Energie entspricht.

$$H(P, i) = -\log d(P) - \log d(i|P)$$

Die Hamiltonian Dynamiken  $\hat{P}$  und  $\hat{i}$

$$\begin{aligned}
\hat{P} &= \frac{\partial}{\partial i} H(P, i) \\
&= -\frac{\partial}{\partial i} \log d(i|P) \\
\\
\hat{i} &= -\frac{\partial}{\partial P} H(P, i) \\
&= \frac{\partial}{\partial P} \log d(P) + \frac{\partial}{\partial P} \log d(i|P)
\end{aligned} \tag{2.2}$$

erhalten die gesamte Energie bzw. Wahrscheinlichkeit und so kann beim HMC Sampling stets ein Schritt berechnet werden, der akzeptiert wird [1], sodass die Markov-Kette auch aus stark gekrümmten Bereichen akzeptierte Stichproben ziehen kann. Besonders in Modellen mit vielen Dimensionen schafft das HMC Sampling so wichtige Effizienzvorteile.

Theoretisch ist HMC Sampling unempfindlich bezüglich stark korrelierter Parameter, allerdings werden die Differenzialgleichungen in der Praxis mittels endlicher Schrittweite gelöst. Das kann dazu führen, dass stark gekrümmte Bereiche der Ziel-Dichte doch nicht mehr erreicht werden. Außerdem gilt es die Schrittweite entsprechend der Parameter-Maßeinheiten und -Größen anzupassen [1].

## 2.4. Stan und R

*Stan* ist eine Open-Source Plattform für statistische Modellierung und high-performance Berechnungen. *Stan* ist für alle in der Datenanalyse weit verbreiteten Sprachen (R, Python, shell, MATLAB, Julia, Stata) verfügbar und läuft auf den gängigen Betriebssystem (Linux, Mac, Windows) [4].

*Stan* ist eine höhere Programmiersprache und ermöglicht die Beschreibung der gemeinsamen Verteilung eines Modells auf einer hohen Abstraktionsebene. Die Sprache unterstützt viele Inferenz-Algorithmen, insbesondere Algorithmen, die das Gradientenabstiegsverfahren nutzen, um eine maximale a-posteriori-Schätzung zu erhalten, HMC Sampling sowie Bayesche Gradientenvariation (stochastic gradient variational Bayes). *Stan* Code wird in C++ kompiliert und die hohe Abstraktionsebene der Sprache bietet dem Nutzer einige Vorteile, z.B. müssen bei der Nutzung von HMC Sampling die Gradienten nicht manuell implementiert werden. Benötigte Gradienten werden built-in mittels C++ Library für automatisches Differenzieren berechnet [1].

Ein *Stan*-Programm muss mindestens einen data-, einen Parameter-Block sowie einen model-Block enthalten. Ein Beispiel dafür ist der *Stan*-Code des in dieser Arbeit verwendeten GARCH-Modells:

Im data-Block wird definiert, welche Daten benötigt werden, um das Modell zu fitten. Die Variablen in diesem Block entsprechen also den tatsächlichen Beobachtungen und der Block definiert, in welchem Format diese Daten vorliegen.

Hier wird zum Beispiel ein Vektor mit Zeitpunkten  $T$ , ein Vektor mit Angabe, ob es sich um eine tatsächliche Beobachtung oder um einen zu vorhersagenden Wert handelt, oder einen Vektor der beobachtete Returns in jedem Zeitpunkt enthält, erwartet.

Im transformed data Block können berechnete Daten/Variablen definiert werden. Hier zum Beispiel wird über die Daten iteriert und festgestellt, wie viele Einträge als *Missing Value* markiert wurden. So werden Beobachtungen und Vorhersagen bzw. die zu vorhersagende Daten unterschieden.

```

1 data {
2   int<lower=0> T;
3   int<lower=0, upper=1> miss_mask[T];
4   real ret_obs[T]; // Note: Masked indices treated as missing
5 }
6 transformed data {
7   int N = 0; // number of missing values
8   for (t in 1:T)
9     if (miss_mask[t] == 1) N = N + 1;
10 }

```

Daran anschließend kommt der parameters-Block, der die unbekannten Parameter des Modells definiert. Anschließend daran kann der transformed parameters-Block dazu genutzt werden, um z.B. die Berechnungsvorschrift weiterer Parameter festzulegen.

Hier wird der Return für jeden Zeitpunkt  $T$  berechnet: Handelt es sich um eine tatsächliche Beobachtung, wird der beobachtete Return genommen, ansonsten erfolgt die Berechnung mit Hilfe der Standardabweichung im aktuellen Zeitpunkt berechnet:

$$\text{ret}[t] = \mu + \sigma[t] * \text{eps\_miss}[t]$$

```

1 parameters {
2   real mu;
3   real<lower=0> alpha0;
4   real<lower=0,upper=1> alpha1;
5   real<lower=0,upper=(1-alpha1)> beta1;
6   real<lower=0> sigma1;
7   real eps_miss[N]; // missing normalized return innovations
8 }
9 transformed parameters {
10   real ret[T]; // returns: observed or r_t=mu+sigma_t*eps_t
11   real<lower=0> sigma[T];
12
13   {
14     int idx = 1; // missing value index

```

```

15
16   sigma[1] = sigma1;
17   if (miss_mask[1] == 1) {
18     ret[1] = mu + sigma[1] * eps_miss[idx];
19     idx = idx + 1;
20   } else
21     ret[1] = ret_obs[1];
22
23   for (t in 2:T) {
24     sigma[t] = sqrt(alpha0
25                   + alpha1 * pow(ret[t - 1] - mu, 2)
26                   + beta1 * pow(sigma[t - 1], 2));
27     if (miss_mask[t] == 1) {
28       ret[t] = mu + sigma[t] * eps_miss[idx];
29       idx = idx + 1;
30     } else
31       ret[t] = ret_obs[t];
32   }
33 }
34 }

```

Zum Schluss kommt der `model`-Block, der die Berechnungsvorschrift der Wahrscheinlichkeitsdichte des Modells enthält [4].

Hier die Annahme, dass  $\mu$  und  $\sigma$  standardnormalverteilt sind. Die Returns sind normalverteilt zu den Parametern des Modells ( $\mu$ ,  $\sigma$ ) wobei es sich, um die Standardabweichung in jedem Zeitpunkt  $T$  handelt. Weitere Erläuterungen befinden sich im Kapitel 3.1

Anschließend daran können im `generated quantities`-Block weitere Berechnungen definiert werden. Hier z.B. die Berechnungsvorschrift der bedingten gemeinsamen Wahrscheinlichkeitsdichte.

```

1  ,
2  model {
3    mu ~ normal(0, 1);
4    sigma1 ~ normal(0, 1);
5
6    ret ~ normal(mu, sigma);
7    // Jacobian correction for transformed innovations
8    for (t in 1:T) {
9      if (miss_mask[t] == 1)
10        target += log(sigma[t]);
11    }
12 }
13 generated quantities {
14   real log_lik[T];
15
16   for (t in 1:T)
17     log_lik[t] = normal_lpdf(ret_obs[t] | mu, sigma[t]);
18 }

```

Der *Stan*-Code der Modelle, die in dieser Arbeit genutzt wurden befindet sich im Anhang und wird im Kapitel 3 näher erläutert.

Die Modelle wurden mittels *Rstan* in *R* gefittet. *Rstan* ermöglicht es *Stan*-Modelle in *R* zu kompilieren, zu testen und zu analysieren. Außerdem wurden die Pakete *tidybayes* und *tidyverse* genutzt, um die Plots zu generieren. Die tidy\* Pakete zeichnen sich dadurch aus, dass sie der komplexen Datenstruktur des berechneten Fits, Informationen leicht entnehmen können und in einer Datenstruktur zur Verfügung stellen, die anschließend z.B. mittels *ggplot*-Packet leicht in *R* visualisiert werden kann.



### 3. Die Modelle

In den letzten Jahren haben agentenbasierte Asset-Pricing-Modelle an Bedeutung gewonnen. Im Gegensatz zur Rationalitätsannahme des Homo oeconomicus, werden nur einfache heuristische Strategien über das Verhalten von Marktteilnehmern getroffen.

Gemeinsamkeiten: readme.txt -> log returns, input, parameter, ....  
stan data und parameter block beschreiben - dann pro kapitel nur die abweichenden Code lines zeigen  
jacobian correction erklären  
agent-based + GARCH als Vergleich

#### 3.1. Generalized Autoregressive Conditional Heteroscedasticity (GARCH)

Generalisierte autoregressive bedingte Heteroskedastizitäts-Modelle sind stochastische Modelle zur Zeitreihenanalyse.

Autoregressive Modelle beschreiben Beobachtungen als Kombination linearer Abhängigkeit von sich selbst sowie einer Zufallsvariable z. B. zufällige Modellfehler. Die Zufallsvariable ist entsprechend nur unter Unsicherheit vorhersagbar. Mit Hilfe von autoregressiven Modellen wird also untersucht inwiefern Beobachtungen in einem Zeitpunkt von der Vergangenheit abhängen.

Autoregressive bedingte Heteroskedastizitäts-Modelle kurz ARCH-Modelle wurden erstmals von Robert F. Engle im Jahr 1982 beschrieben. ARCH( $i$ )-Modelle treffen die Annahme, dass die bedingte Varianz der zufälligen Modellfehler im Zeitpunkt  $t$  vom realisierten Zufallsfehler der letzten  $i$  Vorperioden abhängt. Das heißt, dass große bzw. kleine Fehler dazu neigen in Clustern aufzutreten [5].

Heteroskedastizität ist ein Begriff der Statistik, der die Streuung innerhalb eines Datensatzes beschreibt. Heteroskedastizität liegt in einer Zeitreihe dann vor, wenn die Abweichung der Daten von einer Trendgeraden im Zeitverlauf zu bzw. abnimmt oder auch einfach nicht in jedem Zeitpunkt gleich ist (vgl. Abbildung 3.1). Im Gegensatz dazu bedeutet Homoskedastizität entsprechend, dass die Abweichung der Daten von der Trendgeraden in jedem Zeitpunkt gleich ist.

Eine Verallgemeinerung der ARCH-Modelle wurde 1986 von Tim Bollerslev in Form des GARCH( $i, j$ )-Modells beschrieben. Hier hängt die bedingte Varianz der zufälligen Modellfehler im Zeitpunkt  $t$  nicht nur von Zufallsfehler der letzten  $i$  Vorperioden ab, sondern auch von sich selbst, also der bedingten Varianz der  $j$  vorherigen Zeitpunkte [3].

Sei  $d_t$  eine Zeitreihe und  $\epsilon_t$  das Residuum im Zeitpunkt  $t$ , dann gilt

$$\begin{aligned} d_t &= \sigma_t \cdot \epsilon_t \\ \sigma_t^2 &= a_0 + a_1 d_{t-1}^2 + \dots + a_i d_{t-i}^2 + b_1 \sigma_{t-1}^2 + \dots + b_j \sigma_{t-j}^2 \end{aligned} \quad (3.1)$$




Abbildung 3.1.: Die Daten weisen im Zeitverlauf unterschiedliche Abweichung zur Trendgerade auf, d.h. die Residuen/Störterme sind nicht konstant im Zeitverlauf und es liegt Heteroskedastizität vor (Eigene Darstellung).

Wobei  $a_l, b_k \in \mathbb{R}_{>0}$  mit  $a_i \neq 0$  und  $b_j \neq 0$  gilt. Außerdem sind die Fehlerterme unabhängige identisch verteilte Zufallsvariablen mit  $E(\epsilon_t) = 0$  und  $VAR(\epsilon_t) = 1$ . Somit gilt, dass die bedingte Varianz

$$\sigma_t^2 = VAR(d_t | d_{t-1}, d_{t-2}, \dots)$$

sowohl von Werten der eigenen Vergangenheit als auch von der Vergangenheit der gesamten Zeitreihe abhängt.

In der Ökonometrie werden autoregressive Modelle z.B. für die Modellierung und Prognose von Finanzmarktdaten genutzt. Kurse an Finanzmärkten unterliegen in der Regel keiner gleichmäßigen Volatilität. Phasen mit geringen Schwankungen alternieren mit Phasen starker Schwankungen. Die Zeitreihenanalyse und insbesondere GARCH-Modelle dienen der Identifikation und Beschreibung dieser Volatilitätscluster.

In dieser Arbeit wurde das GARCH(1,1)-Modell als Vergleichsmodell und Baseline berücksichtigt, da es ein in der Ökonometrie weitverbreitetes und genutztes Modell ist.

Beim GARCH(1,1)-Modell hängt die bedingte Varianz nur vom vorherigen Zeitpunkt ab und somit gilt

$$\sigma_t^2 = VAR(d_t | d_{t-1})$$

Entsprechend kann die GARCH(1,1) Zeitreihe wie folgt modelliert werden:

$$\begin{aligned} d_t &= \sigma_t \cdot \epsilon_t \\ \sigma_t^2 &= a_0 + a_1 d_{t-1}^2 + b_1 \sigma_{t-1}^2 \end{aligned} \quad (3.2)$$

Bei den untersuchten Zeitreihen in dieser Arbeit handelt es sich um Kursdaten des S&P 500 Indexes (vgl. Kapitel 4.1), d.h. das GARCH(1,1)-Modell untersucht Volatilitätscluster dieser Preise. Die Preise wurden in Erträge  $r_t$  umgerechnet, wobei je nach Markierung der Beobachtung der beobachtete Ertrag oder der geschätzte Ertrag  $r_t$  verwendet wird.

$$r_t = \mu + \sigma_t \cdot \epsilon_t$$

Entsprechend kann das GARCH(1,1)-Modell zur Modellierung der Volatilität in Form der Standardabweichung der Erträge implementiert werden.

$$\begin{aligned}
 \sigma_t^2 &= a_0 + a_1 d_{t-1}^2 + b_1 \sigma_{t-1}^2 \\
 \sigma_t^2 &= a_0 + a_1 (\sigma_t \cdot \epsilon_t)^2 + b_1 \sigma_{t-1}^2 \\
 \sigma_t^2 &= a_1 (\sigma_{t-1} \cdot \epsilon_{t-1} + \mu - \mu)^2 + b_1 \sigma_{t-1}^2 \\
 \sigma_t^2 &= a_0 + a_1 (r_{t-1} - \mu)^2 + b_1 \sigma_{t-1}^2
 \end{aligned} \tag{3.3}$$

Die Berechnung wird entsprechend im *Stan*-Code als Parameter implementiert.

```

1
2 parameters {
3   real mu;
4   real<lower=0> alpha0;
5   real<lower=0,upper=1> alpha1;
6   real<lower=0,upper=(1-alpha1)> beta1;
7   real<lower=0> sigma1;
8   real eps_miss[N]; // missing normalized return innovations
9 }
10 transformed parameters {
11   real ret[T]; // returns ... observed or r_t = mu + sigma_t * eps_t
12   real<lower=0> sigma[T];
13
14   {
15     int idx = 1; // missing value index
16
17     sigma[1] = sigma1;
18     if (miss_mask[1] == 1) {
19       ret[1] = mu + sigma[1] * eps_miss[idx];
20       idx = idx + 1;
21     } else
22       ret[1] = ret_obs[1];
23
24     for (t in 2:T) {
25       sigma[t] = sqrt(alpha0
26                     + alpha1 * pow(ret[t - 1] - mu, 2)
27                     + beta1 * pow(sigma[t - 1], 2));
28       if (miss_mask[t] == 1) {
29         ret[t] = mu + sigma[t] * eps_miss[idx];
30         idx = idx + 1;
31       } else
32         ret[t] = ret_obs[t];
33     }
34   }
35 }

```

Der Erwartungswert  $\mu$  sowie die Standardabweichung  $\sigma$  sind standardnormalverteilt. Die Erträge folgen einer Normalverteilung mit den Parametern  $\mu$  und  $\sigma$ . Außerdem muss die Ziel-Dichte die Jacobi-Korrektur (*engl. Jacobian Corrector*) für alle Vorhersagen berücksichti-

gen. Schließlich wird im `model`-block des *Stan*-Codes, die a-posteriori-Dichte als normalisierte bedingte Verteilung der Erträge `ret_obs` von  $\mu$  und  $\sigma$  definiert.

```

1 model {
2   mu ~ normal(0, 1);
3   sigma1 ~ normal(0, 1);
4
5   ret ~ normal(mu, sigma);
6   // Jacobian correction for transformed innovations
7   for (t in 1:T) {
8     if (miss_mask[t] == 1)
9       target += log(sigma[t]);
10  }
11 }
12 generated quantities {
13   real log_lik[T];
14
15   for (t in 1:T)
16     log_lik[t] = normal_lpdf(ret_obs[t] | mu, sigma[t]);
17 }

```

Der vollständige *Stan*-Code des GARCH(1,1)-Modells befindet sich im Anhang A.

## 3.2. Franke & Westerhoff (FW)

R. Franke und F. Westerhoff haben ein Modell entwickelt, dass den Markt durch zwei Agentengruppen bzw. Handelsstrategien beschreibt: Fundamental<sup>1</sup>- und Chartist<sup>2</sup>-Trader. Händler können täglich die Strategie ändern, da im Modell ein Mechanismus modelliert wurde, der Herdenverhalten abbildet [6]

Der Anteil der Fundamental-Trader zum Zeitpunkt  $t$  sei

$$n_t^f \in [0, 1]$$

Dann gilt für den Anteil der Chartist-Trader entsprechend

$$n_t^c = 1 - n_t^f$$

da nur zwei Handelsstrategien am Markt existieren[1, 6].

---

<sup>1</sup>**Fundamental-Trader** treffen Kauf bzw. Verkaufsentscheidung auf Basis vorhandener betriebswirtschaftlicher- und Kapitalmarktinformationen wie z.B. Jahresabschlüsse, Inflation, politische Informationen etc.. Für Fundamental-Trader gilt die Annahme, dass Märkte vorhersagbar sind und auf bestimmte Vorfälle vorhersagbares Verhalten zeigen. Durch Informationen über eben diese Vorfälle, kann der Händler informierte Vorhersagen über die Preise am Finanzmarkt treffen.

<sup>2</sup>**Chartist-Trader** treffen Handelsentscheidungen auf Basis historischer Preise und Volumen des Marktes. Chartist-Trader analysieren graphische Repräsentationen der historischen Daten und mathematische Indikatoren, um ihre Handelsstrategie zu optimieren.

Der logarithmierte Preis  $p_t$  im Zeitpunkt  $t$  hängt vom vorherigen Preis sowie der vorherigen durchschnittlichen Nachfrage aller Händler  $d_{t-1}^f$  und  $d_{t-1}^c$  ab.

$$p_t = p_{t-1} + \mu(n_{t-1}^f \cdot d_{t-1}^f + n_{t-1}^c \cdot d_{t-1}^c) \quad (3.4)$$

Fundamental-Trader reagieren auf fehlerhafte Preise z.B. wenn der Marktpreis vom (bekannten) Fundamentalpreis<sup>3</sup>  $p^*$  des Vermögenswertes abweicht. Chartist-Trader reagieren im Gegensatz dazu auf Preisbewegungen der Vergangenheit z.B.  $p_t - p_{t-1}$ . Die Nachfrage selbst ist keine beobachtbare Größe und nur ihre gewichtete Summe beeinflusst die Preise. Sie wird von [1, 6] mittels folgender Dynamiken definiert:

$$\begin{aligned} d_t^f &= f(p^* - p_t) + \epsilon_t^f & \text{mit } \epsilon_t^f &\sim \mathcal{N}(0, \sigma_f^2) \\ d_t^c &= c(p_t - p_{t-1}) + \epsilon_t^c & \text{mit } \epsilon_t^c &\sim \mathcal{N}(0, \sigma_c^2) \end{aligned}$$

Allgemein gilt für zwei unabhängige Zufallsvariablen  $X, Y$ , dass die Verteilung ihrer Summe der Faltung der Verteilung der beiden Zufallsvariablen entspricht.

Sei

$$\begin{aligned} X &\sim \mathcal{N}(\mu_X, \sigma_X^2) \\ Y &\sim \mathcal{N}(\mu_Y, \sigma_Y^2) \end{aligned}$$

dann gilt

$$X + Y \sim \mathcal{N}(\mu_X + \mu_Y, \sigma_X^2 + \sigma_Y^2)$$

So ist es möglich, dass die Nachfrage nicht tatsächlich berechnet bzw. geschätzt werden muss, sondern indirekt durch die Verteilung der kombinierte Nachfrage der existierenden Trader-Gruppen Einfluss auf die logarithmierten Preise  $r_t = p_t - p_{t-1}$  nimmt. Für das stochastische Modell der logarithmierten Preise

$$r_t \sim \mathcal{N}(\mu_t, \sigma_t^2)$$

gelten dann die Verteilungsparameter

$$\begin{aligned} \mu_t &= E[n_{t-1}^f d_{t-1}^f + n_{t-1}^c d_{t-1}^c] \\ &= \mu(n_{t-1}^f f(p^* - p_{t-1}) + n_{t-1}^c c(p_{t-1} - p_{t-2})) \end{aligned} \quad (3.5)$$

und

$$\sigma_t^2 = \mu^2((n_{t-1}^f)^2 \sigma_f^2 + (n_{t-1}^c)^2 \sigma_c^2) \quad (3.6)$$

Die Volatilität  $\sigma_t$  hängt somit von der Größe der einzelnen Trader-Gruppen ab und ändert sich mit der Zeitreihe. Um den Wechsel der Händler von einer Gruppe zur Anderen zu beschreiben, wird zu jedem Zeitpunkt  $t$  deren Anteil am Gesamtmarkt  $n_t^f$  bzw.  $n_t^c$  neu berechnet.

---

<sup>3</sup>Der **Fundamentalpreis** ist der Preis, der rein auf Basis vorhandener Informationen berechnet wurde

Der Aktualisierungs-Quotient entspricht der DCA<sup>4</sup>-HPM<sup>5</sup> Spezifikation nach [1, 6].

Sei  $a_t$  die relative Attraktivität der Fundamental-Strategie gegenüber der Chartist-Strategie. Nach [1, 6] wird die Attraktivität mittels drei beeinflussender Faktoren modelliert (vgl. Formel 3.7). Zum einen wird eine allgemeine Veranlagung (General Predisposition)  $\alpha_0$  und zum anderen Herdenverhalten (Herding)  $\alpha_n$  und fehlerhafte Preise  $\alpha_p$  modelliert (HPM). Da nur eine diskrete Anzahl an Wahlmöglichkeiten bestehen (DCA), ist die Wahrscheinlichkeitsdichte des Modells leicht differenzierbar. Das hat zur Folge, dass ein HMC-Algorithmus leichter Stichproben aus der a-posterior-Verteilung ziehen kann.

Wie bereits erwähnt gilt

$$n_t^c = 1 - n_t^f$$

wobei die Veränderung der Anzahl der Händler zu jedem Zeitpunkt  $t$  durch folgenden Quotienten berechnet wird

$$n_t^f = \frac{1}{1 + e^{-\beta \cdot a_{t-1}}} \quad \beta = 1 \text{ nach [1]}$$

mit

$$a_t = \alpha_0 + \alpha_n(n_t^f - n_t^c) + \alpha_p(p^* - p_t)^2 \quad a_n, a_p > 0 \quad (3.7)$$

Die Parameter  $\beta$  und  $\mu$  sind redundante Parameter werden konstant gehalten, da sie nur die Größenordnung der Parameter  $\alpha_0, \alpha_n, \alpha_p, f, g, \sigma_f$  und  $\sigma_c$  beeinflussen.

Es gilt

$$\beta = 1$$

$$\mu = 0.01$$

Außerdem wird angenommen, dass der Fundamentalpreis unbekannt und nicht beobachtbar ist, das wird durch den logarithmierte Fundamentalpreis  $p^* = 0$  modelliert [1, 6]. Bei der Simulation des Modells auf realen Aktienmarktpreisen bzw. -erträgen, wird aufgrund des unbekannten Fundamentalpreises angenommen, dass dieser sich im Zeitverlauf der geometrischen Brownschen Bewegung<sup>6</sup> folgt.

$$p^* \sim \mathcal{N}(p_{t-1}^*, \sigma_*^2) \quad (3.8)$$

Die Volatilität  $\sigma_t$  in Formel 3.6 hängt über den Marktanteil der Fundamental-Trader  $n_{t-1}^f$  von der Attraktivität  $a_{t-2}$  in Formel 3.7, die wiederum vom logarithmierten Fundamentalpreis  $p_{t-2}^*$  beeinflusst wird. In der vorliegenden Arbeit wurden zwei verschiedene Varianten des Modells nach [1] auf Marktpreisen simuliert. Die Varianten unterscheiden sich in der Berechnung des logarithmierten Fundamentalpreises  $p_t^*$ :

---

<sup>4</sup>Discrete Choice Approach(DCA) nach Formel ?? [1]

<sup>5</sup>Herding, Predisposition, Missalignment(HPM) Spezifikation für Attraktivität der Fundamental-Strategie gegenüber der Chartist-Strategie[1].

<sup>6</sup>Die Brownsche Bewegung ermöglicht die Simulation multiplikativer unabhängiger Zuwächse des Fundamentalpreises [7]

1. Modellierung von  $p_t^*$  durch die Brownsche Bewegung bzw. eine Re-Parametrierung und die Berechnung mittels zufälliger Irrfahrt (Random Walk)
2. Modellierung von  $p_t^*$  durch einen gleitenden Durchschnitt (Moving Average)

In der ersten Variante enthält  $p_t^*$  durch die Brownsche Bewegung nun eine stochastische Komponente. Es handelt sich also um ein stochastisches Volatilitätsmodell [1, 6]. Der vollständige *Stan*-Code befindet sich im Anhang A.

Durch eine nicht zentrierte Umparametrierung nach [1] kann die Performance des HCM-Sampler verbessert werden. Dazu wird  $p_t^*$  mittels  $\epsilon_t^*$  als transformierter Parameter innerhalb einer zufälligen Irrfahrt (Random Walk) berechnet.

$$p_t^* = p_{t-1}^* + \sigma_* \cdot \epsilon_t^* \quad \text{wobei } \epsilon_t^* \sim \mathcal{N}(0, 1) \quad (3.9)$$

Das Modell wird bleibt dadurch unverändert, denn Formel 3.8 und 3.9 sind äquivalent [1].

In Variante 2 wird die Veränderung des logarithmierten Fundamentalpreises durch einen gleitenden Durchschnitt modelliert. Es wird angenommen, dass der Fundamentalpreis von aktuellen sowie vergangenen Werten eines unbekannten stochastischen Terms (Zufallsfehler) abhängt. Auch hier gilt somit, dass es sich um ein stochastisches Volatilitätsmodell handelt [1]. random walk - treedepth 128 ?!

### 3.3. AL herd walk (AL)

tbd

## 4. Simulationen

tbd

### 4.1. Daten & Vorhersagen

S&P 500 data in USD from finance.yahoo:

- daily prices - calculated into return and finally into log return as models expect log return as inputs

- exporting data with dates like Jan 1 2000 to Jan 1 2008 yahoo automatically uses last working day at stock exchange

- always used 30 Predictions - inaccurate as days per month change

=> this all leads to inaccuracy which is ok as the goal is to see the tendency in which way predictions are shifting


monthly prediction during a year in which a major crisis happend: bank crisis 2008 & dotcom 2000?

used 8 years to predict the year 2008 on a monthly basis: e.g. data from Jan 1 2000 to Jan 1 2008 used to predict Jan 2008

### 4.2. Ergebnisse

tbd





# Anhang

## A. Stan Code

### GARCH-Modell

```
1 data {
2   int<lower=0> T;
3   int<lower=0, upper=1> miss_mask[T];
4   real ret_obs[T]; // Note: Masked indices will be treated as missing;
5 }
6 transformed data {
7   int N = 0; // number of missing values
8   for (t in 1:T)
9     if (miss_mask[t] == 1) N = N + 1;
10 }
11 parameters {
12   real mu;
13   real<lower=0> alpha0;
14   real<lower=0, upper=1> alpha1;
15   real<lower=0, upper=(1-alpha1)> beta1;
16   real<lower=0> sigma1;
17   real eps_miss[N]; // missing normalized return innovations
18 }
19 transformed parameters {
20   real ret[T]; // returns ... observed or r_t = mu + sigma_t * eps_t
21   real<lower=0> sigma[T];
22
23   {
24     int idx = 1; // missing value index
25
26     sigma[1] = sigma1;
27     if (miss_mask[1] == 1) {
28       ret[1] = mu + sigma[1] * eps_miss[idx];
29       idx = idx + 1;
30     } else
31       ret[1] = ret_obs[1];
32
33     for (t in 2:T) {
34       sigma[t] = sqrt(alpha0
35                     + alpha1 * pow(ret[t - 1] - mu, 2)
36                     + beta1 * pow(sigma[t - 1], 2));
37       if (miss_mask[t] == 1) {
38         ret[t] = mu + sigma[t] * eps_miss[idx];
39         idx = idx + 1;
40       } else
```

```

41     ret[t] = ret_obs[t];
42   }
43 }
44 }
45 model {
46   mu ~ normal(0, 1);
47   sigma1 ~ normal(0, 1);
48
49   ret ~ normal(mu, sigma);
50   // Jacobian correction for transformed innovations
51   for (t in 1:T) {
52     if (miss_mask[t] == 1)
53       target += log(sigma[t]);
54   }
55 }
56 generated quantities {
57   real log_lik[T];
58
59   for (t in 1:T)
60     log_lik[t] = normal_lpdf(ret_obs[t] | mu, sigma[t]);
61 }

```

## Modell von Franke & Westerhoff (Random Walk)

```

1  ,
2  data {
3    int<lower=0> T; // time points (equally spaced)
4    int<lower=0, upper=1> miss_mask[T];
5    vector[T] ret_obs; // Note: Masked indices will be treated as missing;
6  }
7  transformed data {
8    int N = 0; // number of missing values
9    real ret_sd = sqrt(variance(ret_obs));
10   // mu and beta fixed ... redundant anyways
11   real mu = 0.01;
12   real beta = 1.0;
13
14   for (t in 1:T)
15     if (miss_mask[t] == 1) N = N + 1;
16 }
17 parameters {
18   real<lower=0> phi;
19   real<lower=0> xi;
20   real alpha_0;
21   real<lower=0> alpha_n;
22   real<lower=0> alpha_p;
23   real<lower=0> sigma_f;
24   real<lower=sigma_f> sigma_c;
25   real<lower=0, upper=1> n_f_1;
26   // p_star random walk in non-centered parameterization
27   vector[T] epsilon_star;

```

```

27   real<lower=0> sigma_p_star;
28   vector[N] eps_miss; // missing normalized return innovations
29 }
30 transformed parameters {
31   vector[T] n_f;
32   vector[T] demand;
33   vector[T] sigma;
34   // Note: All prices are actually log prices!
35   vector[T] p_star;
36   vector[T] p;
37   real ret[T];
38
39   p[1] = 0; // wlog log p_1 = 0
40   p_star[1] = p[1] + epsilon_star[1]; // fixme ... interpretation epsilon_raw[1]
41
42   n_f[1] = n_f_1;
43   demand[1] = 0;
44   sigma[1] = mu * sqrt( square(n_f[1] * sigma_f)
45     + square((1 - n_f[1]) * sigma_c));
46   {
47     int idx = 1;
48
49     if (miss_mask[1] == 1) {
50       ret[1] = sigma[1] * eps_miss[idx];
51       idx = idx + 1;
52     } else
53       ret[1] = ret_obs[1];
54
55     for (t in 2:T) {
56       // Note: index shift between prices and returns
57       p[t] = p[t - 1] + ret[t - 1];
58       p_star[t] = p_star[t-1] + sigma_p_star * epsilon_star[t];
59
60       {
61         // equation (HPM)
62         real a = alpha_n * (n_f[t-1] - (1 - n_f[t-1]))
63           + alpha_0
64           + alpha_p * square(p[t-1] - p_star[t-1]);
65         // equation (DCA)
66         n_f[t] = inv_logit(beta * a);
67         demand[t] = mu * (n_f[t] * phi * (p_star[t] - p[t])
68           + (1 - n_f[t]) * xi * (p[t] - p[t-1]));
69         // structured stochastic volatility
70         sigma[t] = mu * sqrt( square(n_f[t] * sigma_f)
71           + square((1 - n_f[t]) * sigma_c));
72       }
73
74       if (miss_mask[t] == 1) {
75         ret[t] = sigma[t] * eps_miss[idx];
76         idx = idx + 1;
77       } else
78         ret[t] = ret_obs[t];

```

```

79   }
80   }
81 }
82 model {
83   phi ~ student_t(5, 0, 1);
84   xi ~ student_t(5, 0, 1);
85   alpha_0 ~ student_t(5, 0, 1);
86   alpha_n ~ student_t(5, 0, 1);
87   alpha_p ~ student_t(5, 0, 1);
88   sigma_f ~ normal(0, ret_sd / mu);
89   sigma_c ~ normal(0, 2.0 * ret_sd / mu);
90   epsilon_star ~ normal(0, 1);
91   sigma_p_star ~ normal(0, ret_sd / 2.0);
92
93   // Price likelihood
94   ret ~ normal(demand, sigma);
95   // Jacobian correction for transformed innovations
96   for (t in 1:T) {
97     if (miss_mask[t] == 1)
98       target += log(sigma[t]);
99   }
100 }
101 generated quantities {
102   vector[T] log_lik;
103
104   for (t in 1:T)
105     log_lik[t] = normal_lpdf(ret_obs[t] | 0, sigma[t]);
106 }

```

## Modell von Franke & Westerhoff (Moving Average)

```

1  ,
2  data {
3    int<lower=0> T; // time points (equally spaced)
4    int<lower=0, upper=1> miss_mask[T];
5    vector[T] ret_obs; // Note: Masked indices will be treated as missing;
6  }
7  transformed data {
8    int N = 0; // number of missing values
9    real ret_sd = sqrt(variance(ret_obs));
10   // mu and beta fixed ... redundant anyways
11   real mu = 0.01;
12   real beta = 1.0;
13
14   for (t in 1:T)
15     if (miss_mask[t] == 1) N = N + 1;
16 }
17 parameters {
18   real<lower=0> phi;
19   real<lower=0> xi;
20   real alpha_0;

```

```

20  real<lower=0> alpha_n;
21  real<lower=0> alpha_p;
22  real<lower=0> sigma_f;
23  real<lower=0> sigma_c;
24  real<lower=0, upper=1> n_f_1;
25  real<lower=0> lenscale_raw;
26  real p_star_0;
27  vector[N] eps_miss; // missing normalized return innovations
28 }
29 transformed parameters {
30   vector[T] n_f;
31   vector[T] demand;
32   vector[T] sigma;
33   // Note: All prices are actually log prices!
34   vector[T] p_star;
35   vector[T] p;
36   real ret[T];
37   real<lower=0> lenscale = 1000 * lenscale_raw;
38   real<lower=0, upper=1> tau = exp( - 1 / lenscale);
39
40   p[1] = 0; // wlog log p_1 = 0
41   p_star[1] = log_mix(tau, p_star_0, p[1]);
42
43   n_f[1] = n_f_1;
44   demand[1] = 0;
45   sigma[1] = mu * sqrt( square(n_f[1] * sigma_f)
46     + square((1 - n_f[1]) * sigma_c));
47   {
48     int idx = 1;
49
50     if (miss_mask[1] == 1) {
51       ret[1] = sigma[1] * eps_miss[idx];
52       idx = idx + 1;
53     } else
54       ret[1] = ret_obs[1];
55
56     for (t in 2:T) {
57       // Note: index shift between prices and returns
58       p[t] = p[t - 1] + ret[t - 1];
59       p_star[t] = log_mix(tau, p_star[t-1], p[t]);
60
61       {
62         // equation (HPM)
63         real a = alpha_n * (n_f[t-1] - (1 - n_f[t-1]))
64           + alpha_0
65           + alpha_p * square(p[t-1] - p_star[t-1]);
66         // equation (DCA)
67         n_f[t] = inv_logit(beta * a);
68         demand[t] = mu * (n_f[t] * phi * (p_star[t] - p[t])
69           + (1 - n_f[t]) * xi * (p[t] - p[t-1]));
70         // structured stochastic volatility
71         sigma[t] = mu * sqrt( square(n_f[t] * sigma_f)

```

```

72         + square((1 - n_f[t]) * sigma_c));
73     }
74
75     if (miss_mask[t] == 1) {
76         ret[t] = sigma[t] * eps_miss[idx];
77         idx = idx + 1;
78     } else
79         ret[t] = ret_obs[t];
80     }
81 }
82 }
83 model {
84     phi ~ student_t(5, 0, 1);
85     xi ~ student_t(5, 0, 1);
86     alpha_0 ~ student_t(5, 0, 1);
87     alpha_n ~ student_t(5, 0, 1);
88     alpha_p ~ student_t(5, 0, 1);
89     sigma_f ~ normal(0, ret_sd / mu);
90     sigma_c ~ normal(0, 2.0 * ret_sd / mu);
91     p_star_0 ~ normal(0, 0.2);
92     lenscale_raw ~ inv_gamma(2, 1); // avoid lower boundary ... lenscale ~ inv_gamma(2, 1000)
93
94     // Price likelihood
95     ret ~ normal(demand, sigma);
96     // Jacobian correction for transformed innovations
97     for (t in 1:T) {
98         if (miss_mask[t] == 1)
99             target += log(sigma[t]);
100     }
101 }
102 generated quantities {
103     vector[T] log_lik;
104
105     for (t in 1:T)
106         log_lik[t] = normal_lpdf(ret_obs[t] | 0, sigma[t]);
107 }

```

## B. R Code

```

1 data {
2     int<lower=0> T;
3     int<lower=0, upper=1> miss_mask[T];
4     real ret_obs[T]; // Note: Masked indices treated as missing
5 }
6 transformed data {
7     int N = 0; // number of missing values
8     for (t in 1:T)
9         if (miss_mask[t] == 1) N = N + 1;
10 }

```

## Literatur

- [1] N. Bertschinger, I. Mozzhorin und S. Sinha. „Reality-check for Econophysics: Likelihood-based fitting of physics-inspired market models to empirical data“. In: *CoRR* abs/1803.03861 (März 2018). arXiv: 1803.03861. URL: <http://arxiv.org/abs/1803.03861>.
- [2] M. Betancourt. „A conceptual introduction to Hamiltonian Monte Carlo“. In: *arXiv pre-print arXiv:1701.02434* (Jan. 2017).
- [3] Tim Bollerslev. „Generalized autoregressive conditional heteroskedasticity“. In: *Journal of econometrics* 31.3 (1986), S. 307–327.
- [4] B. Carpenter, A. Gelman, M. D. Hoffman, D. Lee, B. Goodrich, M. Betancourt, M. Brubaker, J. Guo, P. Li und A. Riddell. „Stan : A Probabilistic Programming Language“. In: *Journal of Statistical Software* 76.1 (Jan. 2017). DOI: 10.18637/jss.v076.i01.
- [5] Robert F Engle. „Autoregressive conditional heteroscedasticity with estimates of the variance of United Kingdom inflation“. In: *Econometrica: Journal of the Econometric Society* (1982), S. 987–1007.
- [6] Reiner Franke und Frank Westerhoff. „Estimation of a structural stochastic volatility model of asset pricing“. In: *Computational Economics* 38.1 (2011), S. 53–83.
- [7] *Geometrische brownsche Bewegung*. Jan. 2018. URL: [https://de.wikipedia.org/wiki/Geometrische\\_brownsche\\_Bewegung](https://de.wikipedia.org/wiki/Geometrische_brownsche_Bewegung).
- [8] K. M. Hanson. „Markov chain Monte Carlo posterior sampling with the Hamiltonian method“. In: Bd. 4322. 2001, S. 4322 - 4322 –12. DOI: 10.1117/12.431119. URL: <https://doi.org/10.1117/12.431119>.
- [9] *Modus*. Juli 2018. URL: [https://de.wikipedia.org/wiki/Modus\\_\(Statistik\)](https://de.wikipedia.org/wiki/Modus_(Statistik)).
- [10] *Simulated Annealing*. Jan. 2019. URL: [https://de.wikipedia.org/wiki/Simulated\\_Annealing](https://de.wikipedia.org/wiki/Simulated_Annealing).
- [11] *Typical set*. Nov. 2018. URL: [https://en.wikipedia.org/wiki/Typical\\_set](https://en.wikipedia.org/wiki/Typical_set).



# Eigenständigkeitserklärung

gemäß Bachelor-Ordnung Informatik 2011 §25 Abs. 11

Hiermit erkläre ich Frau

---

Die vorliegende Arbeit habe ich selbständig und ohne Benutzung anderer als der angegebenen Quellen und Hilfsmittel verfasst.

Frankfurt am Main, den \_\_\_\_\_

---

Unterschrift